

Санкт-Петербургский политехнический университет Петра Великого
Институт компьютерных наук и технологий
Кафедра компьютерных систем и программных технологий

Низкоуровневое программирование

Отчет по лабораторной работе №1
EDSAC

Работу
выполнил:
Кечин В.В.
Группа:
3530901/90004
Преподаватель:
Алексюк А.О.

Санкт-Петербург
2021

Содержание

1. Цель работы	3
2. Программа работы	3
3. Теория	3
4. Ход выполнения работы	4
5. Выводы	9

1. Цель работы

- Разработать программу для EDSAC, реализующую сортировку выбором массива чисел in-place, и предполагающую загрузчик Initial Orders 1. Массив (массивы) данных и другие параметры (преобразуемое число, длина массива, параметр статистики и пр.) располагаются в памяти по фиксированным адресам.
- Выделить определенную вариантом задания функциональность в замкнутую (closed) подпрограмму, разработать вызывающую ее тестовую программу. Использовать возможности загрузчика Initial Orders 2. Адрес обрабатываемого массива данных и другие параметры передавать через ячейки памяти с фиксированными адресами.

2. Программа работы

- Установить симулятор EDSAC
- Изучить и понять принцип работы EDSAC
- Разработать требуемые программы для EDSAC в соответствии с заданным вариантом

3. Теория

Initial Orders 1 – простая программа, которая решает следующую задачу: чтение символов (т.е., разумеется, кодов символов) с перфоленты, формирование кодов инструкций (и констант) и запись их в память. Initial Orders 1 состоит из инструкций, которые загружаются в ячейки 0 – 30 и начинают исполняться после нажатия кнопки «Start». Программа, записанная на ленте, загружается в следующие друг за другом ячейки памяти, начиная с адреса 31. Первая инструкция (загружаемая в ячейку 31) обязательно должна иметь вид $T <N+1> S$, где N – адрес ячейки, в которую загружается последняя инструкция программы. Так, «пустая» программа состоит из 1 инструкции: $T \ 32 \ S$ (результатом ее выполнения будет запись значения аккумулятора в ячейку 32).

Initial Orders 2 – значительно более сложная (семантически) программа, чем Initial Orders 1, имеющая, тем не менее, немногим больше инструкций (41 вместо 31), и решающая схожую задачу. Форма записи инструкции при использовании Initial Orders 2 незначительно отличается от формы записи инструкций в случае Initial Orders 1. Однако, в отличие от Initial Orders 1, при использовании Initial Orders 2 на ленте находятся не только инструкции (или константы) загружаемой программы, но и управляющие последовательности (“control combinations” в терминах Initial Order 2, современный термин – “директивы”), определяющее поведение самой программы Initial Order 2 в процессе загрузки. Директивы Initial Order 2 записываются в той же форме, что и инструкции загружаемой программы. Initial Orders 2 состоит из 41 инструкции, которые загружаются в ячейки 0 – 40 и начинают исполняться после нажатия кнопки «Start». Ячейки 41 – 43 являются для Initial Orders 2 рабочими. По умолчанию, программа, записанная на ленте, загружается в следующие друг за другом ячейки памяти, начиная с адреса 44. Однако, в отличие от Initial Orders 1, адрес загрузки в любой момент может быть изменен с помощью соответствующих директив.

4. Ход выполнения работы

Для удобного написания программы реализуем небольшие программы на языке Python для индексирования программ для загрузчиков IO1 и IO2. Общая идея программы для IO1: Вводим длину массива, адрес его первого элемента и сам массив. Далее в цикле до тех пор, пока у нас есть необработанные элементы, входим во внутренний цикл, в котором ищем максимальный элемент из оставшихся. После меняем максимальный элемент с последним необработанным. Листинг программы IO1:

```
1 [31] T 136[N+1] S
2
3 [32] X0S
4
5 [33] A 122 [len] S
6 [34] A 123 [1] S
7 [35] T 121 [cur_len] S
8
9 [36] A 102 [sw1] S
10 [37] A 125 [f_el] S
11 [38] A 122 [len] S
12 [39] A 122 [len] S
13 [40] S 124 [2] S
14 [41] T 102 [sw1] S [set last_el]
15
16 [42] A 106 [sw2] S
17 [43] A 125 [f_el] S
18 [44] A 122 [len] S
19 [45] A 122 [len] S
20 [46] S 124 [2] S
21 [47] T 106 [sw2] S [set last_el]
22
23 [48] A 71 [for_max] S
24 [49] A 125 [f_el] S
25 [50] T 71 [for_max] S [set 1 el]
26
27 [51] A 0[f_el] S [fl]
28 [52] U 0[f_el] S [tfl]
29
30 [53] A 51 [fl] S
31 [54] A 125 [f_el] S
32 [55] T 51 [fl] S
33
34 [56] A 52 [tfl] S
35 [57] A 125 [f_el] S
36 [58] T 52 [tfl] S указатели[ на первый элемент]
37
38 [loop]
39
40 [59] T 0 S [l1]
41
42 [60] A 121 [cur_len] S
43 [61] S 123 [1] S
44 [62] T 121 [cur_len] S [-1 количество необработанных элементов]
45
46 [63] A 51 [fl] S
47 [64] U 86 [cur] S
48 [65] T 100 [ss1] S
49
50 [66] A 52 [tfl] S
```

```

51 [67] U 87 [cur_d] S
52 [68] T 103 [ss2] S      возвращаемся[ к первому элементу массива для поиска
    ↳ максимума]
53
54 [69] A 123 [1] S
55 [70] T 120 [count2] S   счетчик[ второго цикла = 1]
56
57 [71] A 0 F [for_max]
58 [72] T 118 [max] S      первый[ элемент = макс]
59
60 [loop2]
61 [73] T 0 S [12]        обнуление[ асс]
62
63 [74] A 120 [count2] S
64 [75] S 121 [cur_len] S
65 [76] E 99 [111] S      выход[ из цикла]
66 [77] A 121 [cur_len] S
67 [78] A 123 [1] S
68 [79] T 120 [count2] S   счетчик[2+=1]
69
70 [80] A 86 [cur] S       итерация[ по элементам]
71 [81] A 124 [2] S       итерация[ по элементам]
72 [82] T 86 [cur] S       итерация[ по элементам]
73 [83] A 87 [cur_d] S     итерация[ по элементам]
74 [84] A 124 [2] S       итерация[ по элементам]
75 [85] T 87 [cur_d] S     итерация[ по элементам]
76 [86] A 0 S [cur]        итерация[ по элементам]
77 [87] U 0 S [cur_d]      итерация[ по элементам]
78
79 [88] S 118 [max] S      [-max]
80 [89] G 73 [12] S        возврат[ к первой строчке второго цикла если макс не
    ↳ поменялся]
81 [90] A 118 [max] S      [+max, получаем число, которое было]
82 [91] T 118 [max] S      присваиваем[ max значение]
83
84 [92] A 86 [cur] S
85 [93] T 100 [ss1] S      присваиваем[ max значение для swap]
86
87 [94] A 87 [cur_d] S
88 [95] T 103 [ss2] S      присваиваем[ max значение для swap]
89
90 [96] A 120 [count2] S
91 [97] S 121 [cur_len] S
92 [98] G 73 [12] S        возврат[ к первой строчке второго цикла если не пройдены все
    ↳ значения]
93
94 [swap]
95
96 [99] T 0 S [111]        [асс=0]
97 [100] A 0 S [ss1] загрузка[ в аккумулятор значения из ячейки с максимальным значением
    ↳ ]
98 [101] T 0 S            запись[ этого значения в рабочую ячейку, обнуление аккумулятора]
99 [102] A 0 S [sw1] загрузка[ в аккумулятор значения из последней ячейки]
100 [103] U 0 S [ss2] [ запись этого значения в ячейку с максимальным значением, обнуление
    ↳ аккумулятора]
101 [104] T 3 S
102 [105] A 0 S            загрузка[ в аккумулятор значения из ячейки 0]
103 [106] T 0 S [sw2] запись[ этого значения в последнюю, обнуление аккумулятора]
104
105 [107] A 102 [sw1] S

```

```

106 [108] S 124 [2] S
107 [109] T 102 [sw1] S
108
109 [110] A 106 [sw2] S
110 [111] S 124 [2] S
111 [112] T 106 [sw2] S      убираем[ обработанный элемент из дальнейшей обработки]
112
113 [113] A 119 [count] S
114 [114] A 123 [1] S
115 [115] U 119 [count] S
116 [116] S 122 [len] S
117 [117] G 59 [11] S      конец[ цикла или возврат к первому действию если не прошло
      ↪ итераций = длине]
118
119 [vars]
120 [118] P 0 S [max]
121 [119] P 0 L [count] [= 1]
122 [120] P 0 L [count2] [= 1]
123 [121] P 0 S [cur_len] [= 11]
124 [122] P 5 S [len] [=10]
125 [123] P 0 L [1]
126 [124] P 1 S [2]
127 [125] P 126 [<f_el>] S [f_el]
128
129 [array]
130 [126] P 0 S [<f_el>] [0]
131 [127] P 2 L [5]
132 [128] P 1 S [2]
133 [129] P 3 S [6]
134 [130] P 4 L [9]
135 [131] P 1 L [3]
136 [132] P 0 L [1]
137 [133] P 2 S [4]
138 [134] P 4 S [8]
139 [135] P 3 L [7]

```

Теперь сделаем эту программу подпрограммой - будем ее вызывать из другого места в коде. Для этого добавим установку параметров в начальное положение (для многократного вызова), а также воспользуемся инструкциями загрузчика IO2.

```

1 T 56 K [ директива IO2, установка адреса загрузки ]
2 G K [ директива IO2, фиксация начального адреса подпрограммы ]
3 [0] A 3 F [ пролог: формирование кода инструкции возврата в Асс ]
4 [1] T 89 [<ret>] @ [ пролог: запись инструкции возврата ]
5
6 [2] A 5[ len ] F
7 [3] A 94 [1] @
8 [4] T 93 [cur_len] @
9 [5] A 94 [1] @
10 [6] T 91 [count] @
11 [7] A 94 [1] @
12 [8] T 92 [count2] @ [set_params]
13
14 [9] A 69 [sw1] @
15 [10] A 4 F
16 [11] T 69 [sw1] @ [set_last_el]
17
18 [12] A 73 [sw2] @
19 [13] A 4 F
20 [14] T 73 [sw2] @ [set_last_el]
21

```

```

22 [15] A 38 [for_max] @
23 [16] A 1 F
24 [17] T 38 [for_max] @ [set 1 el]
25 [18] A 0[f_el] F [fl]
26 [19] U 0[f_el] F [tfl]
27
28 [20] A 18 [fl] @
29 [21] A 1 F
30 [22] T 18 [fl] @
31
32 [23] A 19 [tfl] @
33 [24] A 1 F
34 [25] T 19 [tfl] @ указатели[ на первый элемент]
35
36 [loop]
37
38 [26] T 0 F [l1]
39
40 [27] A 93 [cur_len] @
41 [28] S 94 [1] @
42 [29] T 93 [cur_len] @ [-1 количество необработанных элементов]
43
44 [30] A 18 [fl] @
45 [31] U 53 [cur] @
46 [32] T 67 [ss1] @
47
48 [33] A 19 [tfl] @
49 [34] U 54 [cur_d] @
50 [35] T 70 [ss2] @ возвращаемся[ к первому элементу массива для поиска
    ↪ максимума]
51
52 [36] A 94 [1] @
53 [37] T 92 [count2] @ счетчик[ второго цикла = 1]
54
55 [38] A 0[f_el] F [for_max]
56 [39] T 90 [max] @ первый[ элемент = макс]
57
58 [loop2]
59
60 [40] T 0 F [l2] обнуление[ асс]
61
62 [41] A 92 [count2] @
63 [42] S 93 [cur_len] @
64 [43] E 66 [l11] @ [check cond]
65 [44] A 93 [cur_len] @
66 [45] A 94 [1] @
67 [46] T 92 [count2] @ счетчик[2+=1]
68
69 [47] A 53 [cur] @ итерация[ по элементам]
70 [48] A 2 F итерация[ по элементам]
71 [49] T 53 [cur] @ итерация[ по элементам]
72 [50] A 54 [cur_d] @ итерация[ по элементам]
73 [51] A 2 F итерация[ по элементам]
74 [52] T 54 [cur_d] @ итерация[ по элементам]
75
76 [53] A 0[f_el] F [cur] итерация[ по элементам]
77 [54] U 0[f_el] F [cur_d] итерация[ по элементам]
78
79 [55] S 90 [max] @ [-max]
80 [56] G 40 [l2] @ возврат[ к первой строчке второго цикла если макс не

```

```

81  [57] A 90 [max] @      [+max, получаем число, которое было]
82  [58] T 90 [max] @      присваиваем[ max значение]
83
84  [59] A 53 [cur] @
85  [60] T 67 [ss1] @      присваиваем[ max значение для swap]
86
87  [61] A 54 [cur_d] @
88  [62] T 70 [ss2] @      присваиваем[ max значение для swap]
89
90  [63] A 92 [count2] @
91  [64] S 93 [cur_len] @
92  [65] G 40 [l2] @      возврат[ к первой строчке второго цикла если не пройдены все
      ↪ значения]
93
94  [swap]
95  [66] T 0 F [l11]      [acc=0]
96  [67] A 0[f_el] F [ss1] загрузка[ в аккумулятор значения из ячейки с max значением]
97  [68] T 0 F      запись[ этого значения в рабочую ячейку, обнуление аккумулятора]
98  [69] A 0[l_el] F [sw1] загрузка[ в аккумулятор значения из последней ячейки]
99  [70] U 0[f_el] F [ss2] запись[ этого значения в ячейку, где было max значение]
100 [71] T 6 F
101 [72] A 0 F      загрузка[ в аккумулятор значения из ячейки 0]
102 [73] T 0[l_el] F [sw2] [ запись этого значения в последнюю ячейку]
103
104
105 [74] A 69 [sw1] @
106 [75] S 2 F
107 [76] T 69 [sw1] @
108
109 [77] A 73 [sw2] @
110 [78] S 2 F
111 [79] T 73 [sw2] @      убираем[ обработанный элемент из дальнейшей обработки]
112
113 [80] A 91 [count] @
114 [81] A 94 [1] @
115 [82] U 91 [count] @
116 [83] S 5[len] F
117 [84] G 26 [l1] @      конец[ цикла или возврат к первому действию если не прошло
      ↪ итераций = длине]
118
119 [85] T 0 F [ обнуление аккумулятора ] [exit:]
120
121 [86] T 1 F
122 [87] T 4 F
123 [88] T 5 F      [reset]
124
125 [89] E 0 F [<ret>] [ эпилог: инструкция возврата из подпрограммы ]
126
127 [90] P 0 F [max]
128 [91] P 0 F [count] [= 0]
129 [92] P 0 F [count2] [= 0]
130 [93] P 0 F [cur_len] [= ...]
131 [94] P 0 D [1]
132
133 G K      директива[ IO2фиксация, начального адреса программы ]
134 [0] X 0 F [ для пошаговой отладки использовать Z 0 F ]
135 [1] A 14 [<f_el>] @ [ адрес 1 элемента ]
136 [2] T 1 F [ запись адреса 1 эл ячейку 1, обнуление аккумулятора ]
137 [3] A 14 [<f_el>] @ [ адрес последнего элемента ]

```



```

138 [4] A 13 [<len>] @
139 [5] A 13 [<len>] @
140 [6] S 2 F
141 [7] T 4 F [ запись адреса посл эл в ячейку 2, обнуление аккумулятора ]
142 [8] A 13 [<len>] @ [ длина массива ]
143 [9] T 5 F [ запись длины массива в ячейку 3, обнуление аккумулятора ]
144 [10] A 10 вызов[] @ вызов[]
145 [11] G 56 [<sub>] F [/ подпрограммы ]
146 [12] Z 0 F [ останов ]
147
148 [13] P 5 D [<len>] [=11]
149 [14] P 15 [<f_el1>] @ [<f_el>]
150
151 [array:]
152 [15] P 4 F [<f_el1>] [ 8 ]
153 [16] P 0 D [ 1 ]
154 [17] P 1 F [ 2 ]
155 [18] P 1 D [ 3 ]
156 [19] P 5 F [ 10 ]
157 [20] P 2 D [ 5 ]
158 [21] P 3 F [ 6 ]
159 [22] P 4 D [ 9 ]
160 [23] P 0 F [ 0 ]
161 [24] P 3 D [ 7 ]
162 [25] P 2 F [ 4 ]
163 EZ PF директива[ IO2, переход к исполнению]

```

Обе программы реализуют заданный функционал, в чем можно удостовериться, запустив их в симуляторе EDSAC.

5. Выводы

Реализованы программы для загрузчиков IO1 и IO2, которые реализуют сортировку выбором массива чисел in-place.